応用複素関数レポート課題3

桂田 祐史

2023年7月4日,2023年7月18日

1 全般的なこと

- FreeFem++ のインストールが確実に出来るか分からないため、2つの課題 (3A, 3B) を出すかもしれない (FreeFem++ なしでも出来るように)、と言っていましたが、見通しが立ったので、3A のみにします。
- 〆切は7月31日 (月曜) 22:00 です (注: 定期試験は 7/28 に終了する予定)。
- 提出方法: Oh-o! Meiji に A4 サイズの PDF で提出して下さい。
  容量制限 (1 ファイル 30MB) に引っかかった場合は、複数のファイルに分割するなど工夫して下さい。
- 使用するプログラミング言語は、自分の MacBook で実行して見せることが可能なものであればなんでも可。
  (と言っていますが、実際上は FreeFem++ となるでしょう。FreeFem++ に関する質問は気軽にして下さい。)
- プログラムとその実行結果、実行するための情報を含めること。プログラムは別ファイルにしても良いです (こちらが調べやすいので)。
- FreeFem++ の使い方については、
  - `FreeFEM-documentation.pdf` (公式ドキュメント)
    https://doc.freefem.org/pdf/FreeFEM-documentation.pdf
  - 「FreeFem++ の紹介」桂田書いた紹介文書 (更新が必要だけれど…)
    https://m-katsurada.sakura.ne.jp/labo/text/welcome-to-freefem/
  - 「FreeFem++ノート」(桂田の自分用メモ)
    https://m-katsurada.sakura.ne.jp/labo/text/freefem-note/
- サンプル・プログラムを追加しておきます (2023/7/18)。

```
curl -O https://m-katsurada.sakura.ne.jp/program/fem/poisson-kikuchi.edp
curl -O https://m-katsurada.sakura.ne.jp/complex2-2023/potential2d-v1.edp
```

`poisson-kikuchi.edp` は、正方形領域における Poisson 方程式の境界値問題を解くプログラムで、「FreeFem++ で菊地 [1] の Poisson 方程式の例題を解く」を見て下さい。これを見ると多角形領域での Potential 問題の解き方が分かるはずです。`potential2d-v1.edp` については、7月18日の授業で説明します (資料も出す予定)。

2　課題3A

2次元非圧縮ポテンシャル流の定常流で、流体の占める領域 Ω と、その境界 $\Gamma=\partial\Omega$ での流速の法線成分 $v_n := \boldsymbol{v}\cdot\boldsymbol{n}$ が分かっている場合に、速度ポテンシャル ϕ、流れ関数 ψ を計算して、等ポテンシャル線、流線、速度場を可視化せよ。領域 Ω と境界値 (流速の法線成分) v_n は、自分で興味のあるもの、自分の都合の良いものを選んで良い (後の注意を読んでおくこと)。

ϕ は、ポテンシャル問題 (ここでは Laplace 方程式の Neumann 境界値問題)

(1) $$\triangle\phi = 0 \quad (\text{in } \Omega)$$

(2) $$\frac{\partial\phi}{\partial\boldsymbol{n}} = v_n \quad (\text{on } \Gamma)$$

の解である。

ポテンシャル問題 (1), (2) を解いて、等ポテンシャル線と速度場 $\boldsymbol{v}$ を求めるサンプル・プログラム `potential2d-v0-kai.edp` を公開してある (2023/7/18 改訂版に切り替えました)。

`potential2d-v0-kai.edp` は、ターミナルで次のようにして入手する

```
curl -O https://m-katsurada.sakura.ne.jp/complex2/potential2d-v0-kai.edp
```

大筋は、Ω と v_n を自分が決めたものにするようにプログラムを書き換えれば良い。自由度は高いので、工夫・遊び心発揮を期待する。

(過去の例では、円を楕円にするような (一見) 安直な選択があったが、そういう人の多くは、$\int_{\partial\Omega} v_n\, d\sigma = 0$ を満たす v_n が見つけられなかったりしていた。)

流線の書き方には色々なやり方がある (一つくらいノーヒントの問を入れておくことにする)。選んだ問題によっては、分かりやすい図が描けるように調整が必要な場合もある。

注意

次の (1) が非常に重要である。例年それを間違えてナンセンスなレポートを提出する人が少なくない。

(1) $v_n := \boldsymbol{v}\cdot\boldsymbol{n}$ は $\int_\Gamma v_n\, d\sigma = 0$ を満たしている必要がある。実際、Gauss の発散定理と非圧縮性の仮定から

$$\int_\Gamma v_n\, d\sigma = \int_\Gamma \boldsymbol{v}\cdot\boldsymbol{n}\, d\sigma = \int_\Omega \operatorname{div}\boldsymbol{v}\, d\boldsymbol{x} = \int_\Omega 0\, d\boldsymbol{x} = 0.$$

サンプルプログラムでは、円盤領域 $\Omega = \{(x,y)\in\mathbb{R}^2 \mid x^2+y^2<1\}$ なので $\boldsymbol{n} = \begin{pmatrix} x \\ y \end{pmatrix}$ (ここは良く考えること). また一様流 $\boldsymbol{v} = \begin{pmatrix} 1 \\ 2 \end{pmatrix}$ であったので、

$$v_n = \boldsymbol{v}\cdot\boldsymbol{n} = \begin{pmatrix} 1 \\ 2 \end{pmatrix}\cdot\begin{pmatrix} x \\ y \end{pmatrix} = x + 2y$$

としてある。($\boldsymbol{v}$ が Ω で定数関数なので) $\operatorname{div}\boldsymbol{v} = 0$ であるから、当然 $\int_\Gamma v_n\, d\sigma = 0$ も成り立つ。(そうでない場合は、きちんと計算して $\int_\Gamma v_n\, d\sigma = 0$ が成り立つことを確かめる必要がある。)

(2) 湧き出しや吸い込み、点渦など、特異点が Ω 内にあるような問題は、この方法では解くことが出来ない。

FAQ (よくされる質問)

- 境界値 v_n を、`if` を用いた場合分けを含む関数としたいが、FreeFem++ がプログラムを受け付けてくれません→ (FreeFem++ ってしょうがないですね。それはさておき) 例えば

  (a) 弱形式には複数の `int1d()` が指定できるので、境界を分割して、その各部分ごとに (場合分けを含まない) 境界値を与えるようにプログラムを書く。

  (b) 例えば `(x>1 && y>0)` のような式は、条件が成り立つならば 1, 成り立たないならば 0 という値を持つので、`if` を使わずに、式だけで場合分けを含む関数が記述できる。

  のような解決策があります。(2023/7/18 追記) サンプル・プログラム `potential2d-v1.edp` を追加しました。

サンプルプログラム `potential2d-v0-kai.edp` 解説

`potential2d-v0-kai.edp`[1]

```
 1 // potential2d-v0-kai.edp
 2 //   http://nalab.mind.meiji.ac.jp/~mk/complex2/potential2d-v0-kai.edp
 3 //   2次元非圧縮ポテンシャル流
 4 //     速度ポテンシャル，速度を求め、等ポテンシャル線，速度場を描く
 5
 6 border Gamma(t=0,2*pi) { x = cos(t); y = sin(t); } // 円盤領域
 7 int m=40;
 8 mesh Th=buildmesh(Gamma(m));
 9 plot(Th, wait=1, ps="Th.eps");
10 // 次の2行は区分1次多項式を使うという意味
11 fespace Vh(Th,P1);
12 Vh phi, v, v1, v2;
13 // 境界条件の設定
14 func Vn=x+2*y;// Ωが単位円で，V=(1,2) のとき V・n=x+2y
15 func Vn2=((x>0&&y>0) || (x<0&&y<0))*(x+2*y); // 右上と左下のみ
16
17 // 速度ポテンシャルφを求める
18 solve Laplace(phi,v) =
19   int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v)) -int1d(Th,Gamma)(Vn2*v);
20 // 平均を0にする（解の一意性がないため、時々生じる大きなズレを消す
21 real mean = int2d(Th)(phi)/int2d(Th)(1);
22 phi = phi - mean;
23 // φの等高線（等ポテンシャル線）を描く
24 plot(phi,ps="contourpotential.eps",wait=1);
25
26 // ベクトル場 (v1,v2)=∇φ を描く（ちょっと雑なやり方）
27 v1=dx(phi); v2=dy(phi);
28 plot([v1,v2],ps="vectorfield.eps",wait=1);
29
30 // 等ポテンシャル線とベクトル場を同時に描く
31 plot([v1,v2],phi,ps="both.eps", wait=1);
```

- 6 行目で領域 Ω の境界 Γ を指定している (それで Ω が決まる)。
- 8 行目で、Γ を m 分割して、Γ の囲む範囲を三角形分割して、それを `mesh` 型の変数 `Th` に代入している。

- 11 行目、有限要素空間 V_h を区分的 1 次関数の空間と定義している。この辺については この講義では説明を省略する (知りたい人は有限要素法のテキストを読んで下さい)。
- 12 行目、ϕ と試験関数 v, 流速ベクトル場 $\boldsymbol{v}$ の成分 v_1, v_2 を、`Vh` の要素とする。
- 14 行目、境界値の設定をしている。ここで $\boldsymbol{v}\cdot\boldsymbol{n} = x + 2y$ とする理由は上の注意 (1) で説明した。
- 15 行目、境界値の設定をしている (説明のため 2 種類用意した)。
- 18〜19 行目、弱形式の定義。ここを修正する必要はあまりない。
- 21〜22 行目、Neumann 境界値問題は定数だけの不定性を持つため (ϕ が解ならば $\phi+$定数も解)、時々大きなズレが生じ、解の描画などで問題が生じる。平均値 $\dfrac{\displaystyle\iint_\Omega \phi(x,y)\,dx\,dy}{\displaystyle\iint_\Omega dx\,dy}$ を引くことで、平均 $=0$ にしておく。

追加サンプルプログラム `potential2d-v1.edp` 解説

やはり円盤領域での問題だが、境界値を次のように変更した。

> 境界である円周のうち $0 \le \theta \le \pi/6$ で $v_n = -2$, $\pi \le \theta \le 3\pi/4$ で $v_n = 1$, その他で $v_n = 0$.

以前は境界全体を `Gamma` としたが、今回は、`Gamma1` $(0 \le \theta \le \pi/6)$, `Gamma2` $(\pi/6 \le \theta \le \pi)$, `Gamma3` $(\pi \le \theta \le 3\pi/4)$, `Gamma4` $(3\pi/40 \le \theta \le 2\pi)$ と 4 つの部分に分けた。

potential2d-v1.edp[2]

```
1  // potential2d-v1.edp
2  //   http://nalab.mind.meiji.ac.jp/~mk/complex2/potential2d-v1.edp
3  //   2次元非圧縮ポテンシャル流
4  //     速度ポテンシャル，速度を求め、等ポテンシャル線，速度場を描く
5
6  border Gamma1(t=0,pi/6)      { x = cos(t); y = sin(t); }
7  border Gamma2(t=pi/6,pi)     { x = cos(t); y = sin(t); }
8  border Gamma3(t=pi,4*pi/3)   { x = cos(t); y = sin(t); }
9  border Gamma4(t=4*pi/3,2*pi) { x = cos(t); y = sin(t); }
10 int m=20;
11 mesh Th=buildmesh(Gamma1(m)+Gamma2(m)+Gamma3(m)+Gamma4(m));
12 plot(Th, wait=1, ps="Th.eps");
13 // 次の2行は区分1次多項式を使うという意味
14 fespace Vh(Th,P1);
15 Vh phi, v, v1, v2;
16 // 境界条件の設定
17 func Vn1=-2.0;                   // Gamma1 では勢いよく注水
18 func Vn3=1.0;                    // Gamma3 からゆるく排水
19 func Vn=(x>0)*(-2.0)+(x<0)*1.0; // x の符号で Gamma1, Gamma3 を区別してまとめる
20
21 // 速度ポテンシャルφを求める
22 solve Laplace(phi,v) =
23   int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v))
24     -int1d(Th,Gamma1)(Vn1*v)-int1d(Th,Gamma3)(Vn3*v);
25 //     -int1d(Th,Gamma1,Gamma3)(Vn*v);
26 // 平均を0にする (解の一意性がないため、時々生じる大きなズレを消す
27 real mean=int2d(Th)(phi)/int2d(Th)(1);
28 phi=phi-mean;
29 // φの等高線 (等ポテンシャル線) を描く
30 plot(phi,ps="contourpotential.eps",wait=1);
31
32 // ベクトル場 (v1,v2)=∇φ を描く (ちょっと雑なやり方)
33 v1=dx(phi); v2=dy(phi);
34 plot([v1,v2],ps="vectorfield.eps",wait=1);
35
36 // 等ポテンシャル線とベクトル場を同時に描く
37 plot([v1,v2],phi,ps="both.eps", wait=1);
```

- 6〜9 行目、境界 $\partial\Omega$ を 4 つのパートに分けている。
- 17〜19 行と 24 行目、境界での積分を 2 項に分けている。
  - Gamma1 では v1 $(=-2)$, Gamma3 では v3 $(=1)$ とした。
  - 代わりに 25 行目を使うと、Gamma1 と Gamma3 で Vn (=(x>0)*(-2.0)+(x<0)*1.0) という境界値を指定することになる。

流れ関数を求める

これは少し難しいので、やらなくて良い。頑張ろうという人向けのヒント。等ポテンシャル線とベクトル場が描ければとりあえず流れの様子が分かるが、流線も描けると嬉しい。2023/6/6 日の授業 (https://m-katsurada.sakura.ne.jp/complex2-2023/AC08_0606_handout.pdf#page=26) で

$$\psi(\boldsymbol{x}) = \int_{C_{\boldsymbol{x}}} \begin{pmatrix} -v \\ u \end{pmatrix} \cdot d\boldsymbol{r} = \int_{C_{\boldsymbol{x}}} \begin{pmatrix} -v \\ u \end{pmatrix} \cdot \boldsymbol{t}\, ds = \int_{C_{\boldsymbol{x}}} \boldsymbol{v} \cdot \boldsymbol{n}\, ds$$

という式を紹介した。領域が Jordan 領域 (1 つの単純閉曲線で囲まれた領域) であれば、定点 $\boldsymbol{a}$ と $C_{\boldsymbol{x}}$ を境界上に選ぶことで、(境界上での $\boldsymbol{v}\cdot\boldsymbol{n}$ は知っていると仮定しているので) 境界上

での ψ の値 $g(\boldsymbol{x}) := \psi(\boldsymbol{x})$ $(x \in \partial\Omega)$ が得られる。それを用いれば

$$\triangle\psi = 0 \quad (\text{in } \Omega), \quad \psi(\boldsymbol{x}) = g(\boldsymbol{x}) \quad (\text{on } \partial\Omega)$$

という Laplace 方程式の Dirichlet 境界値問題を解くことで ψ が得られる。

3　課題 3B

(FreeFem++ で何かトラブルがあったときのため、念のためもう一つ準備する予定。FreeFem++ がすんなり行ったら、やめるかもしれません。→ やめました。)